
Subject: Re: How to eliminate smaller blob from label_region image

Posted by [Wout De Nolf](#) on Tue, 23 Aug 2011 11:18:20 GMT

[View Forum Message](#) <> [Reply to Message](#)

On Mon, 22 Aug 2011 23:30:15 -0700 (PDT), vijay s
<vijayans.in@gmail.com> wrote:

>
> hi,
>
> thanks for the help and that works fine, now i have the required
> blob (only 2 blob, both are separated and adjacent to each other).
> Now i want to merge both and make as a single blob and want to find
> the new center.
>
> thanks

What do you mean by "new center"? I suppose the center of mass of the combination of the two blobs? If you know you must have two blobs, you don't need to remove the small blobs. Just get the blob index of the two largest blobs and calculate the center of mass of the combined blob.

```
; Get image
path=FILEPATH('pollens.jpg',SUBDIR=['examples','demo','demo data'])
READ_JPEG, path, image
```

```
; Get blob indices
b = LABEL_REGION(image gt 160)
```

```
; Get population and members of each blob
h = HISTOGRAM(b, REVERSE_INDICES=r)
```

```
; Blob indices of the two largest blobs
h[0]=0 ; background
ind=(reverse(sort(h)))[0:1] ; two largest
ind=b[r[r[ind]]] ; get blob indices
```

```
; Get the center of mass of the combined blob
b=b eq ind[0] or b eq ind[1]
totalMass = Total(b)
s=size(b,/dim)
xcm = Total( Total(b, 2) * lindgen(s[0]) ) / totalMass
ycm = Total( Total(b, 1) * lindgen(s[1]) ) / totalMass
```

```
; Show result
```

```
loadct,3  
tvsc1,b  
tmp=indgen(11)-5  
plots,xcm,ycm+tmp,/device,color=150  
plots,xcm+tmp,ycm,/device,color=150  
print,'Center of mass: ',xcm,ycm
```
